MATH529L03: Wave equation

In[1]:=
```
? Sound
```

Out[1]=

Symbol

Sound [*primitives*] represents a sound.

Sound [*primitives*, *t*] specifies that the sound should have duration *t*.

Sound [*primitives*, {t_{min}, t_{max}}] specifies that the sound should extend from time t_{min} to time t_{max}.

In[2]:=
```
Sound[{SoundNote["A"]}]
```

Out[2]=



In[3]:=
```
Sound[{SoundNote["C"], SoundNote["G"]}]
```

Out[3]=



In[4]:=
```
Sound[{SoundNote["F"], SoundNote["F"], SoundNote["F"], SoundNote["D"]}]
```

Out[4]=



In[5]:=
```
Sound[{Play[10 Sin[1000 t (1 + t^2)], {t, 0, .2}],
  Play[10 Sin[500 t (1 + t^3)], {t, 0, .5}]}]
```

Out[5]=



In[6]:=
```
freq = 110;
Sound[{Play[Sin[2 π freq t], {t, 0, 1}]}]
```

Out[7]=



In[8]:=
```
nT = 20; (* Number of terms in trigonometric series *)
```

In[9]:=
```
α = Table[n π / L, {n, 1, nT}]
```

Out[9]=

$$\left\{\frac{\pi}{L}, \frac{2\pi}{L}, \frac{3\pi}{L}, \frac{4\pi}{L}, \frac{5\pi}{L}, \frac{6\pi}{L}, \frac{7\pi}{L}, \frac{8\pi}{L}, \frac{9\pi}{L}, \frac{10\pi}{L}, \frac{11\pi}{L}, \frac{12\pi}{L}, \frac{13\pi}{L}, \frac{14\pi}{L}, \frac{15\pi}{L}, \frac{16\pi}{L}, \frac{17\pi}{L}, \frac{18\pi}{L}, \frac{19\pi}{L}, \frac{20\pi}{L}\right\}$$

In[10]:=
```
f[x_, p_, L_] := If[x ≤ p L, 0.02 x / (p L), 0.02 (L - x) / (L - p L)]
```

In[ ]:=
```
Plot[f[x, 0.2, 1.], {x, 0, 1}, AspectRatio → Automatic]
```

Out[ ]=

In[11]:=
```
F[c_] = Integrate[0.02 x / (p L) Sin[c x], {x, 0, p L}] +
  Integrate[0.02 (L - x) / (L - p L) Sin[c x], {x, p L, L}]
```

Out[11]=

$$\frac{0.02\ (-c\ L\ p\ \text{Cos}[c\ L\ p] + \text{Sin}[c\ L\ p])}{c^2\ L\ p} + \frac{0.02\ c\ L\ (1. - 1.\ p)^2\ \text{Cos}[c\ L\ p] + (-0.02 + 0.02\ p)\ \text{Sin}[c\ L] + (0.02 - 0.02\ p)\ \text{Sin}[c\ L\ p]}{c^2\ L\ (-1. + p)^2}$$

In[12]:=
```
A = Chop[Simplify[Table[2/L F[α[[n]]], {n, 1, nT}]] /. p → 0.2]
```

Out[12]=
```
{0.0148888, 0.00602264, 0.00267673, 0.000930548, 0, -0.000413577, -0.000491644,
 -0.000376415, -0.000183812, 0, 0.000123048, 0.000167295, 0.000142548,
 0.0000759631, 0, -0.0000581593, -0.0000833583, -0.0000743535, -0.0000412431, 0}
```

In[13]:=
```
u[x_, t_, a_] = Sum[ A[[n]] Sin[ α[[n]] x ] Cos[a α[[n]] t] , {n, 1, nT}] /. L → 1
```

Out[13]=
```
0.0148888 Cos[a π t] Sin[π x] + 0.00602264 Cos[2 a π t] Sin[2 π x] +
 0.00267673 Cos[3 a π t] Sin[3 π x] + 0.000930548 Cos[4 a π t] Sin[4 π x] -
 0.000413577 Cos[6 a π t] Sin[6 π x] - 0.000491644 Cos[7 a π t] Sin[7 π x] -
 0.000376415 Cos[8 a π t] Sin[8 π x] - 0.000183812 Cos[9 a π t] Sin[9 π x] +
 0.000123048 Cos[11 a π t] Sin[11 π x] + 0.000167295 Cos[12 a π t] Sin[12 π x] +
 0.000142548 Cos[13 a π t] Sin[13 π x] + 0.0000759631 Cos[14 a π t] Sin[14 π x] -
 0.0000581593 Cos[16 a π t] Sin[16 π x] - 0.0000833583 Cos[17 a π t] Sin[17 π x] -
 0.0000743535 Cos[18 a π t] Sin[18 π x] - 0.0000412431 Cos[19 a π t] Sin[19 π x]
```

In[14]:=
```
Plot[Table[u[x, t, 20.], {t, 0, 0.4, 0.02}], {x, 0, 1},
 PlotRange → All, Frame → True, PlotLabel → "Wave equation solution",
 FrameLabel → {"x", "u(x,t)"}, GridLines → Automatic]
```

Out[14]=



In[15]:=
```
Manipulate[Plot[u[x, t, 20.], {x, 0, 1}, PlotRange → {All, {-0.025, 0.025}},
   Frame → True, PlotLabel → "Wave equation solution",
   FrameLabel → {"x", "u(x,t)"}, GridLines → Automatic], {t, 0, 0.4, 0.005}]
```

Out[15]=



In[16]:=
```
u[0.2, t, 20.]
```

Out[16]=
```
0.0087514 Cos[62.8319 t] + 0.00572787 Cos[125.664 t] +
 0.00254572 Cos[188.496 t] + 0.000546963 Cos[251.327 t] +
 0.000243095 Cos[376.991 t] + 0.000467581 Cos[439.823 t] +
 0.000357992 Cos[502.655 t] + 0.000108042 Cos[565.487 t] +
 0.0000723256 Cos[691.15 t] + 0.000159107 Cos[753.982 t] +
 0.000135571 Cos[816.814 t] + 0.00004465 Cos[879.646 t] +
 0.0000341852 Cos[1005.31 t] + 0.0000792784 Cos[1068.14 t] +
 0.0000707144 Cos[1130.97 t] + 0.0000242421 Cos[1193.81 t]
```

In[18]:=
```
Sound[{Play[u[0.2, t, 20.], {t, 0, 0.2}]}]
```

Out[18]=



In[19]:=
```
Sound[{Play[Sum[u[x, t, 800.], {x, 0.1, 0.9, 0.1}], {t, 0, 0.8}]}]
```

Out[19]=



Questions to investigate

Effect of pluck shape upon sound

Make initial shape of plucked string to be :
(1) triangle;
(2) sine wave centered on string;
(3) double sine wave with a node on the middle of the string;
(4) parabolic initial pluck shape.

(1)

In each case produce the sound emitted by the string.